

Pràctiques Integrades

1er de Matemàtiques

Pràctica 14

curs 2002–03

14 Càlcul de derivades i determinació de les característiques del gràfic d'una funció

Amb Maple es pot obtenir fàcilment la derivada d'una expressió **expr** qualsevol. Com ja deveu saber la comanda que fa aquesta feina és **diff(expr,var)** on **expr** és l'expressió que es vol derivar i **var** és la variable respecte la que es farà la derivada.

Exemple 14.1

Podem fer la derivada de l'expressió $x^3 - 3x^2 + 5x + 4$ respecte x amb

```
> diff(x^3-3*x^2+5*x+4,x);
```

i si tenim definida una funció $f(x)$ també podrem fer el mateix

```
> f:= x-> x^3-3*x^2+5*x+4;  
> diff(f(x),x);
```

Teniu en compte, però, que segons com us poseu a fer els càlculs podeu tenir sorpreses. Podeu preveure quin serà el resultat de l'operació següent després d'haver definit la funció **f**?

```
> diff(f,x);
```

Noteu que Maple necessita conèixer respecte quina variable s'ha de fer la derivada ja que en una expressió hi poden haver diferents paràmetres. Per exemple, les expressions

Exemple 14.2

```
> diff(x^2*cos(b)-exp(y^2)*tan(x),x);  
> diff(x^2*cos(b)-exp(y^2)*tan(x),y);  
> diff(x^2*cos(b)-exp(y^2)*tan(x),b);
```

produeixen tres resultats diferents.

Quan treballem amb la comanda **diff** sovint el que ens interessa és definir una nova funció que és la derivada d'una $f(x)$ que ja tenim definida. Com que **diff** treballa amb expressions i dóna com a resultat expressions, aquest procés no és tan immediat com es podria pensar. Si proveu de fer

```

> f:= x-> (x^3-4)/(x^2+3);
> g:= x-> diff(f(x),x);
> g(x);
> g(1);

```

veureu que els resultats semblen no tenir massa lògica.

A part del mecanisme de la comanda `unapply`, que ja hem vist com funciona en una prctica anterior, quan tenim una funció f que depèn d'una variable podem obtenir una nova funció g que és la derivada de la primera amb la comanda `D( )`.

Exemple 14.3

Si definim la funció $f(x) = x^3 - 4x$ i volem tenir la funció $g(x) = f'(x)$ (que òbviament ha de ser $g(x) = 3x^2 - 4$) es pot fer

```

> f:= x-> x^3-4*x;
> g:= D(f);
> g(x);
> g(1);

```

Quan es té definida una funció f que depèn de més d'una variable es pot especificar respecte quina de les variables es vol fer la derivada utilitzant `D[i](f)` on i és un enter que indica que es vol fer la derivada de f respecte la variable que ocupa el lloc i .

Exemple 14.4

Si tenim definida la funció $f(x, y, b) = x^2 \cos(b) - \exp(y^2) \tan(x)$ la primera variable és la x , la segona és la y i la tercera és la b . Per tant, fent

```

> f:= (x,y,b)-> x^2*cos(b)-exp(y^2)*tan(x);
> g1:= D[1](f);
> g2:= D[2](f);
> g3:= D[3](f);

```

obindrem tres funcions $g1$, $g2$, $g3$ que són les derivades de f respecte x , y , b respectivament.

Exercici 14.1

Definiu una funció h que sigui $h(x, y) = x \frac{\exp(xy) \sin(y^2)}{\ln(x^2 + y^2 + 2)}$ i calculeu

- La funció $h1$ derivada de h respecte x .
- La funció $h2$ derivada de h respecte y .
- Els valors de $h1$ i $h2$ per a $x=1$ i $y=\pi/4$.
- La funció $h11$ derivada segona de la funció h respecte x .

És clar que la resposta a l'últim apartat de l'exercici anterior pot ser

```
> h11:= D[1](D[1](h));
```

Com que això no és massa pràctic hi ha dreceres que ens permeten fer el mateix amb més facilitat. Una expressió de la forma $x\$n$ és equivalent a una successió de n còpies de l'expressió x . Per exemple

```
> x$4;
```

produeix x, x, x, x . D'aquesta forma en les comandes `diff` o `D` podem fer

```
> diff(sin(x)-x*(cos(x))^3,x$4);
> D[1$4](cos)(x);
```

14.1 Estudi del comportament del gràfic d'una funció

Un dels problemes més clàssics del càlcul infinitesimal elemental és el de determinar totes les característiques remarcables del comportament d'una funció (asíntotes, creixement, extrems, convexitat, punts d'inflexió). La comanda `plot`, que dibuixa el gràfic d'una funció, ajuda molt a determinar algunes d'aquestes característiques però és clar que la determinació exacta (o amb una bona aproximació) dels punts remarcables no es podrà fer sense solucionar algunes equacions, determinar algun límit i calcular les derivades que calgui. Amb Maple tots els càlculs necessaris es podran fer sense massa problemes i en l'exemple següent podeu veure com fer tot el procés d'una manera sistemàtica.

Exemple 14.5

Considerem la funció $f(x) = \frac{5x}{x^2 - 4}$ i determinem totes les característiques del seu gràfic.

En primer lloc definim la funció i les derivades primera i segona

```
> f:=x-> (5*x)/(x^2-4);
> fp:= D(f);
> fpp:= D(fp);
```

A continuació podem calcular els límits de f a l'infinit (encara que en aquest cas tothom hauria de poder veure el resultat sense Maple)

```
> limit(f(x),x=infinity);
> limit(f(x), x=-infinity);
```

Seguidament, i tenint en compte que l'expressió amb la que estem tractant és una fracció, veiem que hi ha asímptotes verticals ja que els límits

```
> limit(f(x),x=2,right);limit(f(x),x=2,left);
> limit(f(x), x=-2,right);limit(f(x), x=-2,left);
```

són infinits.

El següent pas consistirà en determinar els extrems locals i els intervals de creixement i decreixement de la funció

```
> solve(fp(x));
> solve(fp(x)>0);
> solve(fp(x)<0);
```

Finalment es pot determinar la convexitat i els punts d'inflexió

```
> solve(fpp(x));
> solve(fpp(x)>0);
> solve(fpp(x)<0);
```

Com que qualsevol estudi d'aquest tipus no està complet si no es fa un dibuixet podeu fer el següent exercici:

Exercici 14.2

Feu un gràfic en el que apareixen:

- Les asímptotes de f dibuixades en color verd.
- Els punts d'extrem local de la funció f .
- Els punts d'inflexió del gràfic de f .
- El gràfic de la funció f amb colors diferents d'acord amb la convexitat que tingui (vermell per un tipus de convexitat i marró per l'altre).

Nota: Mireu en l'ajuda de Maple quina opció de `plot` es pot utilitzar per a fer que es tinguin en compte les possibles discontinuïtats a l'hora de dibuixar un gràfic.

Exercici 14.3

Per a practicar una mica més podeu mirar de fer una cosa semblant amb les funcions següents:

$$f(x) = (x^2 - 3x + 2) \exp(x) \quad g(x) = \frac{x^2 - x}{16 - x^2} \quad h(x) = 2(\cos^2 x) - x^2$$

Fent per a cada una d'elles dos gràfics de colors, un on es posi de manifest els intervals de creixement i decreixement i l'altre on es mostri la convexitat.